```
fun append (xs,ys) =
    if xs=[]
    then ys
    else (hd xs)::append(tl xs,ys)

fun map (f,xs) =
    case xs of
       [] => []
       | x::xs' => (f x)::(map(f,xs'))

val a = map (increment, [4,8,12,16])
val b = map (hd, [[8,6],[7,5],[3,0,9]])
```

Programming Languages Dan Grossman

Pattern-Matching for Each-Of Types: The Truth
About Function Arguments

An exciting segment

Learn some deep truths about "what is really going on"

- Using much more syntactic sugar than we realized
- Every val-binding and function-binding uses pattern-matching
- Every function in ML takes exactly one argument

First need to extend our definition of pattern-matching...

Each-of types

So far have used pattern-matching for one of types because we needed a way to access the values

Pattern matching also works for records and tuples:

- The pattern (x1,...,xn)matches the tuple value (v1,...,vn)
- The pattern {f1=x1, ..., fn=xn} matches the record value {f1=v1, ..., fn=vn} (and fields can be reordered)

Example

This is poor style, but based on what I told you so far, the only way to use patterns

Works but poor style to have one-branch cases

```
fun sum_triple triple =
   case triple of
    (x, y, z) => x + y + z

fun full_name r =
   case r of
   {first=x, middle=y, last=z} =>
    x ^ " " ^ y ^ " " ^ z
```

Val-binding patterns

- New feature: A val-binding can use a pattern, not just a variable
 - (Turns out variables are just one kind of pattern, so we just told you a half-truth in lecture 1)

$$val p = e$$

- Great for getting (all) pieces out of an each-of type
 - Can also get only parts out (not shown here)
- Usually poor style to put a constructor pattern in a val-binding
 - Tests for the one variant and raises an exception if a different one is there (like hd, tl, and valOf)

Better example

This is okay style

- Though we will improve it again next
- Semantically identical to one-branch case expressions

```
fun sum_triple triple =
  let val (x, y, z) = triple
  in
       x + y + z
  end

fun full_name r =
  let val {first=x, middle=y, last=z} = r
  in
       x ^ " " ^ y ^ " " ^ z
  end
```

Function-argument patterns

A function argument can also be a pattern

Match against the argument in a function call

$$fun f p = e$$

Examples (great style!):

```
fun sum_triple (x, y, z) =
    x + y + z

fun full_name {first=x, middle=y, last=z} =
    x ^ " " ^ y ^ " " ^ z
```

A new way to go

- For Homework 2:
 - Do not use the # character
 - Do not need to write down any explicit types

Hmm

A function that takes one triple of type int*int*int and returns an int that is their sum:

A function that takes three int arguments and returns an int that is their sum

See the difference? (Me neither.) ©

The truth about functions

- In ML, every function takes exactly one argument (*)
- What we call multi-argument functions are just functions taking one tuple argument, implemented with a tuple pattern in the function binding
 - Elegant and flexible language design
- Enables cute and useful things you cannot do in Java, e.g.,

```
fun rotate_left (x, y, z) = (y, z, x)
fun rotate_right t = rotate_left(rotate_left t)
```

^{* &}quot;Zero arguments" is the unit pattern () matching the unit value ()